
[bookmark: _GoBack]1. Fix the coloring of the section “Stata code”. The correct coloring is shown below.
Stata code:
*Data generation
clear
version 15
set seed 123

*500 leaders (note: “a” is the unobserved effect)
set obs 500
gen lead_n = _n
gen a = rnormal()
gen z = round(25 + 2*rnormal() + .2*a)

*10 raters for each leader
expand 10
bys lead_n: gen rater = _n
gen x = a + rnormal()
gen y = .5*x - .5*z + 5*a + 2*rnormal()

*Data estimation

*1 OLS with Level 1 and Level 2 predictor
reg y x z, vce(cluster lead_n)

*2 OLS with cluster dummies and Level 1 predictor
reg y x i.lead_n, vce(cluster lead_n)

*3 GLS FE with Level 1 predictor
xtset lead_n rater
xtreg y x, fe vce(cluster lead_n)

*4 OLS with Level 1 predictor and its cluster mean 
bys lead_n: egen x_cl = mean(x)
reg y x x_cl, cluster(lead_n)

*5 OLS with Level 1 predictor, its cluster mean, and Level 2 predictor
reg y x x_cl z, cluster(lead_n)

*6 GLS RE with Level 1 predictor, its cluster mean, and Level 2 predictor
xtreg y x x_cl z, cluster(lead_n) robust
test x_cl

*7 MLE RE with Level 1 predictor, its cluster mean, and Level 2 predictor
mixed y x x_cl z || lead_n:
est store one

*compare above estimate with below using likelihood ratio test
mixed y x z || lead_n:
est store two
lrtest one two

*8 ML RE with cluster-mean centered Level 1 predictor and Level 2 predictor
gen x_grp_cen = x - x_cl
mixed y x_grp_cen z || lead_n: 

*9 GLS RE with cluster-mean centered Level 1 predictor and Level 2 predictor
xtreg y x z, cluster(lead_n) robust

*Wald test of the random effects assumption (equivalent to test x_cl of
*model 6
xtoverid

*10 GLS RE with grand-mean centered Level 1 predictor and Level 2 predictor
egen x_grnd_mean = mean(x)
gen x_grnd_cen  = x - x_grnd_mean 

xtreg y x_grnd_cen z, cluster(lead_n)

*11 ML RE with grand-mean centered Level 1 predictor and Level 2 predictor
mixed y x_grp_cen x_cl z || lead_n:


2. Ensure that when copy-pasted from the PDF to a text editor, Stata code and R code retain correct formatting. There are currently at least two problems:
1. Line breaks are lost. For example, copy-pasting
“*Data generation
clear
version 15
set seed 123”
results in 
“*Data generation
clear version 15 set seed 123”
2. = and + and possibly other characters are converted to special characters. For example, copy-pasting
“gen z = round(25 + 2*rnormal() + .2*a)”
produces
“gen z 1⁄4 round(25 þ 2*rnormal() þ .2*a)”
